

GREEN TEA PROGRAMMING LANGUAGE MANUAL

<https://gtlang.com>

Contents

Introduction.....	3
Prerequisites.....	3
Before reading:.....	3
Installation.....	4
Windows.....	4
Linux.....	4
Debian.....	4
Centos.....	4
MacOS.....	4
Usage.....	4
Interprept.....	4
Compile:.....	4
Language file.....	4
Syntax.....	5
Comments.....	5
Keywords.....	6
Commands.....	6
Blocks.....	6
Inclusion.....	7
Variables.....	7
Assignment:.....	8
Get value:.....	8
Scope.....	8
Function.....	8
Declare.....	9
Regex-name function.....	9
Expressions.....	10
Operators.....	11

Arithmetic operators.....	11
Assignment operators.....	11
Comparison operators.....	12
Increment/Decrement operators.....	12
Logical operators.....	12
String operators.....	13
Array operators.....	13
Conditional assignment operators.....	13
Precedence and brackets.....	13
Data types.....	14
Strings.....	15
Number.....	15
Booleans.....	15
Arrays.....	16
Declare:.....	16
Get/assign value for element:.....	16
Objects.....	16
Initialize.....	16
Access object peroperties or method:.....	16
NULL.....	17
Functions.....	17
Conditional structure.....	17
If else structures.....	17
?? expression.....	18
Loop structures.....	19
For structures.....	19
Simple for structures:.....	19
Full for structures:.....	20
For-in with arrays.....	20
Foreach.....	21
While loop structures.....	21
Loop controllers.....	21
Break.....	22
Continue.....	22
Error handlers.....	23

error_handler_function.....	23
try...catch...finally structures.....	24
defcat...deffin structures.....	25
Multi-threading.....	25
File based multi threading.....	25
Function based multi-threading.....	26
Class.....	26
Constructors.....	27
Access modifiers.....	27
Inherit.....	28
Static.....	28
Convert from array to object:.....	29
Generate class from no class object.....	29

Introduction

Green Tea is aimed for people who are new to programming, just like Sketch. It is a simple language that purpose is removing all the things you should remember and let you deploy your idea in Zen mode. "Your ideas matter, not the syntax".

- + Simple and easy to learn, even for people haven't know anything about programming
- + Easy to switch from other languages
- + Could be used to syntactically convert a source code of a language to another.
- + Use your native language to code. Why do you have to learn English fore code?

Green Tea is inspired by PHP, and is a script/compiled language, functional/object-oriented language

Parts of this document is copied from the PHP manual.

Prerequisites

Green Tea is made for newcomers. No prerequisites here.

Before reading

In this document, example code are blue and on-screen in/output are orange.

Code:

```
Hello World
```

```
Input:
```

```
< John
```

```
Output:
```

```
> 10
```

Installation

Windows

Not supported

Linux

Generally, you just need to download the binary, put it somewhere, *chmod +x* it, then add it parent folder to *PATH* environment variable.

Debian

Centos

Not supported

MacOS

Not supported

Usage

Interprept

```
gtlang <sourcefile>
```

Ex:

```
gtlang hello_world.gtc
```

Compile:

```
gtcompiler <sourcefile> <output file>
```

Ex:

```
gtcompiler hello_world.gtc hello_world
```

Language file

Green Tea is the multi-language programming language

To use another language, use `use_language_file` keyword

The variables' name, functions' name, classes' name, keywords could be translated, but hard-coded strings could not.

Ex:

In ru.gtlang, we could define:

```
if : если
for : для
echo : эхо
number : номер
```

```
// (could add more here)
```

in the source file, use

```
use_language_file ru.gtlang
now you could code:
```

```
если $номер=0
    @эхо "$номер ноль"
```

Syntax

Let start with a Hello world program:

Green Tea syntax is inspired by PHP, Shell Script, Javascript, Python, C++, Java, in decreasing order.

```
Hello World
```

```
> Hello World
```

Congratulation. You are now a certificated Green Tea developer, with a lot of years of experiences ☐.

Comments

Comments in code will not be executed

Start with // to the end of line, or between /* and */

```
$a : 3 // create var a and assign 3
```

or

```
/* this is a comment
    create var a and assign 3
end of comment */
```

```
$a := 3
```

You can use html tags in comments.

```
/* <html>this is <b>a comment</b><br />
    create var <b>a</b> and assign 3<br />
end of comment </html>*/
$a := 3
```


Inclusion

You could use include keyword to add another file to current execution when program run.

Ex:

File a.gtc

```
@sum $a, $b
    return $a + $b
@echo "this is file a\n"
File b.gtc
```

```
@echo "begin file b\n"
include "a.gtc"
@echo "this is file b\n"
@echo @sum 3, 4
This is equivalent to:
```

```
@echo "begin file b\n"
@sum $a, $b
    return $a + $b
@echo "this is file a\n"
@echo "this is file b\n"
@echo @sum 3, 4
> begin file b
> this is file a
> this is file b
> 7
```

About the function

```
@sum $a, $b
    return $a + $b
please see function section
```

Variables

Variables (vars) are places in memory when we store a values

Variables in Green Tea are represented by a dollar sign followed by the name of the variable. The variable name is case-sensitive.

Variable names follow the same rules as other labels in GreenTea. A valid variable name starts with a letter or underscore, followed by any number of letters, numbers, or underscores. As a regular expression, it would be expressed thus: `^[a-zA-Z_\x80-\xff][a-zA-Z0-9_\x80-\xff]$`*

Ex: `$var_name`

Assignment

Put a value into a var

```
$var : <value>
```

Ex:

```
$name : John
```

```
$age : 24
```

Or assigning multiple vars at once:

```
$name, $age : John, 24
```

Get value

To get a variable's value, use its name:

```
$name : John
```

```
$name
```

```
> John
```

Scope

A variable defined in a function only usable in that function.

If you want to access global variables inside function, use \$\$ instead of \$

```
$a=0
```

```
@function
```

```
    $b: 1
```

```
    $$c: 2
```

```
    @echo $a // error
```

```
    @echo $$a // output: 0
```

```
@echo $b // error
```

```
@echo $c // output: 2
```

Function

Functions are tasks, type its name, provide parameters and it'll do its task.

Functions in Green Tea are represented by a @ sign followed by the name of the function. The function name is case-sensitive.

Function names follow the same rules as other labels in GreenTea. A valid function name starts with a letter or underscore, followed by any number of letters, numbers, or underscores. As a regular expression, it would be expressed thus: `^[a-zA-Z_\x80-\xff][a-zA-Z0-9_\x80-\xff]$`*

The very first function you should know is @echo. It prints var's value to screen.

```
@echo "Hello world!!!"
```

```
> Hello world!!!
```


Declare

```
@function_name  
    <block>// the things this function do
```

Ex:

```
@my_function  
    @echo 123  
    @echo 456
```

Then you call it

```
@my_function  
> 123456
```

Adding some parameters, separated by commas:

```
@say_hi_to $param_1, $param_2  
    @echo Hello $param_1 I am $param_2  
@say_hi_to Annie John
```

Or:

```
Annie @say_hi_to John  
> Hello Annie I am John
```

Functions could return value, when they do, outer code could get the return value

```
@sum $a, $b  
    return $a + $b  
$sum_total : (@sum 1, 2) + (@sum 3, 4)  
> 10
```

Functions could return multiple values, like Python

```
@sum $a, $b  
    return $a + $b, $a - $b  
$sum, $sub : (@sum 4, 3)  
@echo "$sum and $sub"  
> 7 and 1
```

Note: when using more than 1 function on a line (function and operator, 2 functions,...), you must wrap function call with ()

When you do not have return in your function or have only return with no parameter, function return `false`.

Regex-name function

Let write a function that add 1 to a number, then write it out.

```
@add1 $number  
    @echo ($number + 1)
```

Now, we need function that add 2 to a number, we will need to write new function which nearly exact content, which is boring. We provide Regular expression function to solve this.

```
@<function name's static part>{< function name's regular expression part>}  
<parameters>
```

```
<actions>
```

You could use `$_func_name` to access the value at the regular expression on function name.

Ex:

Let's rewrite the `@addX` function.

```
@add{[0-9]*} $num  
    @echo( $_func_name + $num)  
@add1  2  
> 3  
@add3  4  
> 7
```

In case that a function call match multi Regex-name functions, only the firstly declared Regex-name function matched will run.

Expressions

Expressions includes vars and anything else with value

Ex:

```
1 + 2  
"hello"  
@sqr 2
```

To get the result value from expression, you could use assignment:

```
$value: 1+2+3/3  
@echo "value:" $value  
> value: 4
```

You could also get the value of expression in previous command using `$?`

```
1+2+3/3  
@echo "value:" $?  
> value: 4
```

Operators

Operators are operations on value(s), including:

- Arithmetic operators

- Assignment operators
- Comparison operators
- Increment/Decrement operators
- Logical operators
- String operators
- Array operators
- Conditional assignment operators

Arithmetic operators

Operator	Name	Example	Result
+	Addition	$\$x + \y	Sum of $\$x$ and $\$y$
-	Subtraction	$\$x - \y	Difference of $\$x$ and $\$y$
*	Multiplication	$\$x * \y	Product of $\$x$ and $\$y$
/	Division	$\$x / \y	Quotient of $\$x$ and $\$y$
%	Modulus	$\$x \% \y	Remainder of $\$x$ divided by $\$y$
**	Exponentiation	$\$x ** \y	Result of raising $\$x$ to the $\$y$ 'th power

Assignment operators

The basic assignment operator in GreenTea is ":". It means that the left operand gets set to the value of the assignment expression on the right.

Assignment	Same as...	Description	Example
$\$x : y$		The left operand gets set to the value of the expression on the right	$\$age : 1$
$\$x + : y$	$x : x + y$	Addition	$\$year + : 2$
$\$x - : y$	$x : x - y$	Subtraction	$\$year - : 2$
$\$x * : y$	$x : x * y$	Multiplication	$\$money * : 2$
$\$x / : y$	$x : x / y$	Division	$\$number / : 3$
$x \% : y$	$x : x \% y$	Modulus	$\$count \% : 5$

Assignments operator return assigned value

$\$a : 4$

$\$?$

> 4

Comparison operators

Operator	Name	Example	Result
=	Equal	$\$x = \y	Returns true if $\$x$ is equal to $\$y$
==	Identical	$\$x == \y	Returns true if $\$x$ is equal to $\$y$, and they are of the same type
!=	Not equal	$\$x != \y	Returns true if $\$x$ is not equal to $\$y$
!==	Not	$\$x !==$	Returns true if $\$x$ is not equal to $\$y$, or they are not of the

	identical	\$y	same type
>	Greater than	\$x > \$y	Returns true if \$x is greater than \$y
<	Less than	\$x < \$y	Returns true if \$x is less than \$y
>=	Greater than or equal to	\$x >= \$y	Returns true if \$x is greater than or equal to \$y
<=	Less than or equal to	\$x <= \$y	Returns true if \$x is less than or equal to \$y

Increment/Decrement operators

Operator	Name	Description
++\$x	Pre-increment	Increments \$x by one, then returns \$x
\$x++	Post-increment	Returns \$x, then increments \$x by one
--\$x	Pre-decrement	Decrements \$x by one, then returns \$x
\$x--	Post-decrement	Returns \$x, then decrements \$x by one

Logical operators

Operator	Name	Example	Result
and	And	\$x and \$y	True if both \$x and \$y are true
or	Or	\$x or \$y	True if either \$x or \$y is true
xor	Xor	\$x xor \$y	True if either \$x or \$y is true, but not both
&&	And	\$x && \$y	True if both \$x and \$y are true
	Or	\$x \$y	True if either \$x or \$y is true
!	Not	!\$x	True if \$x is not true

String operators

Operator	Name	Example	Result
+	Concatenation	\$txt1 + \$txt2	Concatenation of \$txt1 and \$txt2
+=	Concatenation assignment	\$txt1 += \$txt2	Appends \$txt2 to \$txt1
/	split	\$string / \$delimiter	Return an array, which is result of splitting \$string with \$delimiter

You could concatenate 2 string by putting them near by:

`ab == $a+$b`

Array operators

Operator	Name	Example	Result
+	Union	<code>\$x + \$y</code>	Union of <code>\$x</code> and <code>\$y</code>
=	Equality	<code>\$x == \$y</code>	Returns true if <code>\$x</code> and <code>\$y</code> have the same key/value pairs
==	Identity	<code>\$x === \$y</code>	Returns true if <code>\$x</code> and <code>\$y</code> have the same key/value pairs in the same order and of the same types
!=	Inequality	<code>\$x != \$y</code>	Returns true if <code>\$x</code> is not equal to <code>\$y</code>
!==	Non-identity	<code>\$x !== \$y</code>	Returns true if <code>\$x</code> is not identical to <code>\$y</code>

Conditional assignment operators

??	Ternary	<code>\$x : expr1 ? expr2 ? expr3</code>	Returns the value of <code>\$x</code> . The value of <code>\$x</code> is <code>expr2</code> if <code>expr1 = TRUE</code> . The value of <code>\$x</code> is <code>expr3</code> if <code>expr1 = FALSE</code>
----	---------	--	--

Precedence and brackets

Expression are calculated by Precedence order. Which operator have higher precedence will be calculated earlier. For example:

$$1 + 2 * 5 = 1 + 10 = 11$$

In order to let the lower precedence to be calculated first, we could put the lower-rank operators and its objects round brackets

$$(1 + 2) * 5 = 3 * 5 = 15$$

The following table lists the operators in order of precedence, with the highest-precedence ones at the top. Operators on the same line have equal precedence, in which case associativity decides grouping.

Associativity	Operators	Additional Information
(n/a)	clone new	clone and new
right	**	arithmetic
(n/a)	+ - ++ -- ~ (int) (float) (string) (array) (object) (bool) @	arithmetic (unary + and -), increment/decrement, bitwise, type casting and error control
left	instanceof	type
(n/a)	!	logical
left	* / %	arithmetic

left	+ - .	arithmetic (binary + and -), array and string (. prior to PHP 8.0.0)
left	<< >>	bitwise
left	+	string
non-associative	< <= > >=	comparison
non-associative	= != == !== <>	comparison
left	&	bitwise and references
left	^	bitwise
left		bitwise
left	&&	logical
left		logical
non-associative	? :	left-associative
right	: += -= *= /" %: &: :	assignment
(n/a)	yield from	yield from
(n/a)	yield	yield
(n/a)	print	print
left	and	logical
left	xor	logical
left	or	logical

Data types

Green Tea support following data types:

- String
- Number
- Boolean
- Array
- Object
- NULL
- Function

String

A string is a sequence of characters

Hello world

"Hello world"

'Hello world'

There are some standard escaped characters included in Green Tea's string:

Sequence	Meaning
\n	linefeed (LF or 0x0A (10) in ASCII)
\r	carriage return (CR or 0x0D (13) in ASCII)
\t	horizontal tab (HT or 0x09 (9) in ASCII)
\v	vertical tab (VT or 0x0B (11) in ASCII)

<code>\e</code>	escape (ESC or 0x1B (27) in ASCII)
<code>\f</code>	form feed (FF or 0x0C (12) in ASCII)
<code>\\</code>	backslash
<code>\\$</code>	dollar sign
<code>\"</code>	double-quote
<code>\[0-7]{1,3}</code>	the sequence of characters matching the regular expression is a character in octal notation, which silently overflows to fit in a byte (e.g. <code>"\400" == "\000"</code>)
<code>\x[0-9A-Fa-f]{1,2}</code>	the sequence of characters matching the regular expression is a character in hexadecimal notation
<code>\u{[0-9A-Fa-f]+}</code>	the sequence of characters matching the regular expression is a Unicode codepoint, which will be output to the string as that codepoint's UTF-8 representation

We recommend you use quotation marks for string until you remember the word that shouldn't be used unquoted.

Number

Numbers include integers and doubles:

-17

3.14

0

0.1

infinite

Boolean

Booleans are either `true` or `false`

Array

An array is a sequence of elements. Elements could be in different types, elements also could be array. An array is a set of key-value pair, contains all unique keys. Keys could be number, string

Types of element's value could be mixed

Declare:

Use `[]` operator to declare value for arrays.

```
$array : [0=>a,1=>b, 2=>c] // with keys
```

```
$array_mixed_type : [1, hello, true, [4.6, false]] // auto key, auto-increase from 0, 1, ...
```

Get/assign value for element:

Use `$array_var_name[key]`

(Using above array declarations)

```

$array[1]
> b
$array[0] : test
$array[0]
> test
$array_mixed_type [3][2]
> false
Or
$array_mixed_type [3,2]
> false
$array_mixed_type [3,2] : true
$array_mixed_type [3][2]
> true
Add new element to array:
$array[]: d

```

You could iterate through array, see section about looping for detailed.

Object

Objects are sets of properties and methods that use the common templates called classes. They will be describe more in object oriented section

Objects are represent of object oriented programming.

Initialize

```
$object : (@new $Class_name)
```

Access object peroperties or method:

use operator . (single dot)

```

$object . (@changeName John)
$object . $name
> John

```

NULL

NULL is the type that hold no value

Function

You could assign a function for a var

```

@function_add $a, $b
    return $a + $b
$var_add : @function_add

```


Conditional structure

Conditional structures are where program choose which part of program to be run by using condition. Includes:

- if else structures
- ? : operators.

If else structures

When you want to split the flow of code into multiple cases base on some conditions. You could use if clause:

```
if <condition>
    <action if condition is satisfied>
else
    <action if condition is not satisfied>
```

Both actions could be single command or block

For example, you write a simple program that test if user input even number:

```
@echo 'Please enter a even integer:'
@read $number_input // read user input into $number_input
if $number_input%2 =0
    @echo This is an even number
    @echo Congratulation
else
    @echo This is not an even number
```

When user run the program and enter 1

```
> Please enter a even integer:
< 1
> This is not an even number
```

When user run the program and enter 2

```
> Please enter a even integer:
< 2
> This is an even number
> Congratulation
```

Multiple if else:

Used when there are multiple cases.

Ex: there are 3 students in class with corresponding student code. Write a program let user input a number, then output their name: 1 (John), 2 (Annie), 3 (Sabrina).

```
@echo "Enter student code 1, 2 or 3"
@read $entered_student_code
if $entered_student_code = 1
```

```

    @echo John
elif $entered_student_code = 2
    echo Annie
elif $entered_student_code = 3
    echo Sabrina
else
    echo Wrong number.

```

?? expression

You can write a short expression that get value based on condition.

<condition>?<value if condition is satisfied>:< value if condition is not satisfied>

Ex: A simple guest game

```

$thinking_number:(@random_int 1, 2) // $a get value 1 or 2
randomly
@echo "I am thinking about a number: 1 or 2. Can you guest
which I am thinking?"
@read $user_input
$thinking_number : $user_input ? "Congratulation, you win" ?
"Better luck next time"
$?

```

In case user guest correctly

```
> I am thinking about a number: 1 or 2. Can you guest which I
am thinking?
```

```
< 1
```

```
> Congratulation, you win
```

Otherwise

```
> I am thinking about a number: 1 or 2. Can you guest which I
am thinking?
```

```
<2
```

```
> Better luck next time
```

Loop structure

Loop structures are used when you want program run a part of it repeatedly, when you don't want to write repeated code or when you don't know exactly the times it loop. Including:

- for structure
- while structure

For structure

For structure usually use when you know how many times you will loop

Simple for structures:

for <time counts> times

<tasks to do repeatedly>

<tasks to do repeatedly> could be single command or block

`$_time` is a var that hold the loop time, starting with **1**.

Ex: Write 3 time "I'll never play games in classes.", with number:

```
for 3 times
    @echo $_time + ". I'll never play games in classes."
> 1. I'll never play games in classes.
> 2. I'll never play games in classes.
> 3. I'll never play games in classes.
```

You could also use `at` keyword here

Ex: Write 4 time "I'll never play games in classes.", at time 3, write "This is time 3":

```
for 3 times
    @echo $_time+". I'll never play games in classes."
    at $_time = 3
        " This is time 3"
```

```
> 1. I'll never play games in classes.
> 2. I'll never play games in classes.
> 3. I'll never play games in classes. This is time 3
```

You could replace `times` with any variable:

```
for 3 $i:
    for 3 $j:
        if $j = $i:
            @echo ' * '
        else:
            @echo '   '
    @echo "\n" //new line
> *
>  *
>  *  *
```

Full for structures:

for <initiation>,<break_condition>,<increasement>

<tasks to do repeatedly>

Ex: Write 3 time "I'll never play games in classes, with number":

```
for $i:1,$i<3,$i++
    @echo ($i+1)". I'll never play games in classes."
> 1. I'll never play games in classes.
> 2. I'll never play games in classes.
> 3. I'll never play games in classes.
```

For-in with arrays

You could iterate through elements in array and do actions for each of those element musing `for` and `in`.

```
for <element's value holding var> in <array>
    <action for each element>
```

In case you need both the key and the value of each element:

```
for < element's key holding var> => <element's value holding var> in <array>
    <action for each element>
```

Ex: write out all name in an array which is a list of names.

```
$list:["Annie","Bob", "John", "Sabrina"]
for $name in $list
    @echo $name + "\n" // \n is the end of output line.
> Annie
> Bob
> John
> Sabrina
>
```

Now, write out all name in an array which is a list of names with it's index.

```
$list : ["Annie","Bob", "John", "Sabrina"]
for $index => $name in $list
    @echo $index + ". " + $name + "\n"
> 1. Annie
> 2. Bob
> 3. John
> 4. Sabrina
>
```

Foreach

```
foreach $list, $k, $v:
```

```
foreach $list, $v:
```

```
foreach $list:
```

```
    $_k, $_v
```

while structure

Use while loop when you couldn't know how many time it loops

```
while <loop condition>
```

```
    <actions to be executed with each loop>
```

When the loop condition's value is true, actions to be executed with each loop will be executed again and again. If the loop condition's value is false, the loop stop, and program continue pass this structure.

Ex: Find all Bob in an array, which is a names list.

```
$list : ["Annie","Bob", "Bob", "Sabrina"]
$i : 1
@echo "Positions of Bob in list:\n"
while $i <= 4
    if $list[$i] = Bob:
        @echo $i + "\n"
    $i++
> Positions of Bob in list:
> 2
> 3
```

Note: You should double check the loop condition to avoid a indefinite loop.

Loop controllers

Loop controller are used to control the execution inside a loop. Includes:

- Break
- Continue

You can use loop controllers to control the loop flexibly.

Break

Break command will break current loop

Ex: : Find first Bob in an array, which is a names list. Since we don't need find continue after 1st Bob, we could use `break` after we found one.

```
$list : ["Annie","Bob", "Bob", "Sabrina"]
$i : 1
@echo "Positions of Bob in list:\n"
while $i <= 4
    if $list[$i] = Bob:
        @echo $i + "\n"
        break
    $i++
```

```
> Positions of Bob in list:
```

```
> 2
```

Bob at position 3 is ignored

You could break outer loop using `break <number>`. With number is the outer-level number of the loop to be break. If you want to break parent loop, use `break 2`; if you want to break grand parent loop, use `break 3`; ...

Ex: Find 1st Bob in an array, which is a names table.

```
$table :[
    ["Jack", "Daniel", "Bill", "Sabrina"],
    ["Annie", "Bob", "John", "William"],
    ["Mike", "Muriel", "Bob", "Brittney"]
]
for $i => $row in $table
    for $j => $name in $row
        if $name = "Bob"
            @echo "Found Bob at row "...
                +($i+1)+...
                ", column "+($j+1)
            break 2
```

```
> Found first Bob at row 2, column 2
```

Continue

To skip the current loop and continue next loop

Ex: Find all Bob in an array, which is a names list.

```
$list : ["Annie","Bob", "Bob", "Sabrina"]
$i : 1
@echo "Positions of Bob in list:\n"
while $i <= 4
    if $list[$i] != Bob
        $i++
        continue
    @echo $i + "\n"
    $i++
```

```
> Positions of Bob in list:
```

```
> 2
```

```
> 3
```

When `$list[$i]` is not *Bob*, `@echo $i + "\n"` won't be executed.

You could also use `continue <number>` to skip the containing outer loops, like with `break`.

Ex: There is an array, each line has only 1 Bob. Find Bob of each line, which is a names table.

```

$stable :[
  ["Jack", "Daniel", "Bill", "Sabrina"],
  ["Annie", "Bob", "John", "William"],
  ["Mike", "Muriel", "Bob", "Brittney"]
]
for $i => $row in $stable
  for $j => $name in $row:
    if $name != "Bob"
      @echo "Found Bob at row "+($i+1)+"...
          ", column "+($j+1)
      continue 2
> Found Bob at row 2, column 2
> Found Bob at row 3, column 3

```

Error handlers

Green Tea provide 2 ways to handle error:

`error_handler_function`

`try...catch...finally` structures

`defcat...deffin` (default catch)

error_handler_function

You could declare this function to handle how to show when unhandled error happened

```

@error_handler_function($errno, $errstr, $errfile, $errline)
  //do somethings

```

`errno`

The first parameter, `errno`, will be passed the level of the error raised, as an integer.

`errstr`

The second parameter, `errstr`, will be passed the error message, as a string.

`errfile`

If the callback accepts a third parameter, `errfile`, it will be passed the filename that the error was raised in, as a string.

`errline`

If the callback accepts a fourth parameter, `errline`, it will be passed the line number where the error was raised, as an integer.

errcontext

If the callback accepts a fifth parameter, `errcontext`, it will be passed an array that points to the active symbol table at the point the error occurred. In other words, `errcontext` will contain an array of every variable that existed in the scope the error was triggered in. User error handlers must not modify the error context.

try...catch...finally structures

Used to catch possible exception in try block. If a Exception match multiple catches, only the 1st catch block matched will run. Finally block run whenever any Exception is catch or not.

Ex: If `^DevidedByZeroException` happened (`$num2=0`), catch block 1 will be triggered). If other exception happened, catch block 2 will run. Finally block run whenever any Exception is catch or not.

```
@devide $num1 $num2
  try
    $result = $num1 / $num2
  catch ^DevidedByZeroException
    // catch block 1
    @echo "devided by zero"
    $result = false
  catch ^Exception
    // catch block 2
    @cho "other exception"
    $result = false
  finally
    return $result
```

defcat...deffin structures

You could use `defcat` as default catch for functions, which will catch Exceptions when running current function

`defcat` and `deffin` should be at the end of function

```
@devide $num1 $num2
  $result = $num1 / $num2
  defcat
    ^DevidedByZeroException
      // catch block 1
      @echo "devided by zero"
      $result = false
    ^Exception
      // catch block 2
      @cho "other exception"
```



```

        $result = false
    deffin
        return $result

```

You could use `defcat` and `deffin` outside of functions, which will handle exception for the whole program. They should be at the end of source files.

Multi-threading

Green Tea support multi threading processing. Includes:

- File-based multi-threading (simple)
- Function-based multi-threading,

Vars will not be shared among threads, you must use parameters and return values to exchange data among threads.

File based multi threading

Use `@create_file_thread` function to create a thread that execute another file. On thread file, you could return value to main thread using `thread_return`. Don't use `exit` in thread file, it will also exit main thread.

File *thread1.gtc* :

```

for 3 times
    @sleep 1 // wait 1s
    "Thread1 - running for "+ $_time + "s"
thread_return $_args[0] // file argument #1

```

File *main.glc*:

```

$thread1 : (@create_file_thread "thread1.gtc Hello")
"Start thread1"
@thread_start $thread1
@thread_wait $thread1
$result : (@thread_result $thread1)
"Thread1 result is: " + $result

```

Run *main.glc*

```

> Start thread1
> Thread1 - running for 1s
> Thread1 - running for 2s
> Thread1 - running for 3s
> Thread1 result is: Hello

```

File based multi threading should have parameters which are numbers or strings, like "Hello" in the above example.

Function based multi-threading

Use `@create_function_thread` function to create a thread that execute a function. On thread file, you could return value to main thread using normal return. Don't use `exit` in thread function, it will also exit main thread.

```
@thread1 $param
  for 3 times
    @sleep 1 // wait 1s
    "Thread1 - running for "+ $_time + "s"
  return $param // file argument #1
$thread1_function : @thread1
$thread1 :(@create_file_thread $thread1_function, "Hello")
"Start thread1"
@thread_start $thread1
@thread_wait $thread1
$result : (@thread_result $thread1)
"Thread1 result is: " + $result
> Start thread1
> Thread1 - running for 1s
> Thread1 - running for 2s
> Thread1 - running for 3s
> Thread1 result is: Hello
```

Function based multi threading could have any kind of parameters.

Class

Classes are "template" to create objects.

Class define methods (functions) and properties (variables) of object.

Class start with `^`

To access object's assets, use dot operator.

Ex:

```
^Human
  $name
  $email
  @print_info
    @echo "$name - $email \n"
```

Then we could create object

```
$people_1 :(@new ^Human)
$people_1.$name: John
$people_1.$email: "john@gmail.com"
$people_1.(@print_info)
> John - john@gmail.com
```

Constructors

Constructors usually add data to object when create a new one, you could create several using **new**. If you don't create a constructor, Green Tea will create a default one. If you have more than 1 constructors, make sure they don't have the same parameters count, since Green Tea will use parameters count to detect which constructor it used.

Ex:

```
^Human
  $name
  $email
  @new $input_name, $input_email
    $name, $email : $input_name, $input_email

  @new $input_name
    $name: $input_name
  @print_info
    @echo "$name - $email \n"
new ^Human, John, "john@gmail.com"
$?.(@print_info)
> John - john@gmail.com
new ^Human, John
$?.(@print_info)
> John -
```

Access modifiers

Green Tea support private and public (default) modifier.

Private asset could not be access outside object

Ex:

```
^Human
  $name
  private $email
  @new $input_name, $input_email
    $name, $email : $input_name, $input_email

  @new $input_name
    $name: $input_name
  @print_info
    @echo "$name - $email \n"
@new ^Human, John, "john@gmail.com"
$?.(@print_info)
> John - john@gmail.com
new ^Human, John, "john@gmail.com"
```

```
@echo $?.$name
> John
new ^Human, John, "john@gmail.com"
@echo $?.$email
(Error - permission denied)
```

Inherit

One class could inherit some other classes using operator << . When doing so, child class could add assets, overload method of the parent class. But could not remove or change type of the parent one's assets.

With class Human above, we could have:

```
^Student << ^Human
  $point
  @print_info
    @echo "$name - $email - $point \n"
$student1: (new ^Student, John, "john@gmail.com")
$student1.$point: 4
$student1.(@print_info)
> John - john@gmail.com - 8
```

Static

Static assets are asset of the class, user don't need to create object to access those static assets.

```
^Human
  static $spicies: Homo Sapien
  static @getSpicies
    return $spicies
^Human.(@getSpicies)
> Homo Sapien
```

Convert from array to object:

An array with all keys satisfied variable's name rules could be converted to object with no class and all public properties.

Ex:

```
$obj: (@to_obj ["name"=> "John", "age"-> 24])
```

Generate class from no class object

An object with no class could generate a class using function @gen_class.

Ex:

```
^Human1 = @gen_class $obj
Now $obj belong to class ^Human1.
```


You are already a programmer. Give us 5 minutes, you will then know green tea programming language.

Variables are \$, functions are @.

```
$number : 3
@echo_hello_world
    @echo "Hello world"
(@echo_hello_world)
> Hello World
```

Green Tea blocks of commands are indent-based. Just like Python, but you can mix spaces and tabs.

```
if $a < 0
    @echo "lesser than 0"
```

Assignments are : and = is comparision.

```
$a:3 // assignment
$a=3 // comparision
```

New for:

```
for 3 times
    @echo $_time + " "
    at $_time = 3
        @echo " (end)"
> 1 2 3 (end)
```

You know the basic. If you have any question, find it on our full document.